

Sección 6: Manejando Funciones

Lab: Uso de Funciones para el manejo de datos.

Información general.

Generalmente los datos no están en la forma o la cantidad en que se requieren: hay que transformarlos para obtener nuevos datos, y las principales herramientas para hacerlo es el uso de Funciones.

Afortunadamente, Excel posee muchas funciones que puedes utilizar para transformar datos y obtener información, tal como la quieres.

Todo lo que necesitas es saber que existen, qué obtienes con ellas, y qué argumentos debes pasarles para que puedan hacer su trabajo.

Ejercicio 1: Modificando datos existentes.

Tarea 1: Concatenando información.

- Abre el archivo Maratón_Inicio.xlsx que está en el directorio de trabajo de la sección (C:\AEY\S06\), y guárdalo con el nombre Maratón.xlsx.
- 2. Asegúrate de estar en la Hoja **Pruebas**.
- 3. El archivo contiene información de las pruebas de maratón olímpicas, desde Atenas 1896, a Londres 2012. Los datos que se tienen son:
 - a. ID del evento, que aún no se tiene.
 - b. La ciudad sede de los juegos.
 - c. Año en que se celebraron los juegos.
 - d. Atleta ganador de la prueba.
 - e. País de origen del ganador de la prueba.
 - f. Marca obtenida en la prueba.
 - g. Tres campos de notas generales.
- 4. El *ID del evento* es un dato que no tenemos, y que se forma por la concatenación del *Año* y la primera letra de *Prueba*, que indica el género de los competidores.
 - a. De esa forma, el *ID del Evento* que debe estar en **A4** es «1896H», que es el año de los juegos, y la primera letra de «Hombres».
 - b. Colócate en A4.
 - c. Introduce la siguiente fórmula:
 - « =CONCATENAR(C4,EXTRAE(D4,1,1))»
 - d. **CONCATENAR()** se encargará de concatenar unir— el valor que está en **C4**, que en este caso es

- «1896», con el resultado de extraer de la Celda **D4**, a partir del primer caracter, un caracter, para lo cual se utiliza la función **EXTRAER()**.
- e. Como sabes, **EXTRAER()** requiere como argumentos la cadena de la cual se extraerá una parte, la posición inicial a partir de la cual hará la extracción, y el número de caracteres que se extraerán.
- f. Haz doble clic en el ícono de Autollenado de la Celda A4, para que se reproduzca la operación con todas las Filas del conjunto de datos.
 - i. Se deberán generar todos los ID de evento.

I	A	1	* :	▼ : X ✓ fx =CONCATENAR(C4,EXTRAE(D4,1,1))			
	4	Α	В	С	D	E	
N	1	Maratón Ol	ímpico				
	2						
	3	ID del Evento	Juegos	Año	Prueba	Atleta	
4	4	1896H	ATENAS	1896	HOMBRES	Spyridon Louis	
	5	1896M	ATENAS	1896	MUJERES		
	6	1900H	PARÍS	1900	HOMBRES	Michel Theato	
	7	1900M	PARÍS	1900	MUJERES		
	8	1904H	SAN LUIS	1904	HOMBRES	Thomas Hicks	
	9	1904M	SAN LUIS	1904	MUJERES		

► Tarea 2: Normalizando el formato de los datos.

- 5. Una de las transformaciones que con más frecuencia te encontrarás, es que se tiene información en formatos no consistentes, y deberás corregir eso.
- 6. Como puedes ver, toda la información textual está en mayúsculas, excepto el nombre del *Atleta*.
- 7. Para darle consistencia al formato, debes convertir a mayúsculas el nombre del *Atleta*; el detalle es que no hay manera de transformar el valor directamente y sobre la


misma Celda que contiene el dato a transformar, por lo que tendrás que utilizar rangos de paso.

- a. Colócate en F4.
- b. Presiona **CTRL+BARRA ESPACIADORA**, para seleccionar la columna.
- c. Presiona CTRL-+, para agregar una Columna nueva.
- d. Se deberá crear una nueva Columna.
- e. Colócate en F4.
- f. Escribe la siguiente función:

«==MAYUSC(E4)»


- g. Como sabes, la Función MAYUSC() retorna el equivalente en mayúsculas de un texto que se proporciona como argumento.
 - i. En este caso, retorna la versión en mayúsculas de **E4**.
- h. Ahora, has doble clic sobre el ícono de **Autollenado** de la celda **F4** para que se haga un autollenado vertical.
- i. Como puedes ver, tienes dos veces el nombre del *Atleta*.
 - i. Pudiera parecer que la opción lógica para dejar los datos como los queremos es eliminar la Columna que tiene los datos en minúsculas.
 - ii. Colócate en E4.
 - iii. Presiona **CTRL+BARRA ESPACIADORA**, para seleccionar la columna.
 - iv. Ahora presiona **CTRL+-**, para eliminar la Columna.

j. Como podrás ver, se ha generado un error. Esto se debe a que los nombres en mayúsculas son el resultado de una función que utiliza los nombres en minúsculas; al quitarlos, la función se queda sin argumentos.



- k. Presiona **CTRL+Z** para deshacer el último cambio, y que todo quede como estaba antes del error.
- 8. Copia los datos usando un rango de paso, y usando pegado especial.
 - a. Selecciona los nombres en mayúscula, que se encuentran en el Rango de celdas **F4:F57**.
 - b. Presiona **CTRL+C**, para copiar el Rango al portapapeles.
 - c. Ahora debes copiar los nombres en mayúsculas sobre los nombres en minúscula.
 - d. Selecciona la Celda E4.
 - e. Utilizando la **Cinta de opciones**, selecciona la opción *INICIO Portapapeles*; haz clic en la parte inferior del ícono *Pegar*, para que funcione como combo y te permita ver más opciones de pegado.

f. Aparecerá un menú que contiene diferentes formas de pegado: Pegado normal, pegado de funciones, pegado de formato, y así. Selecciona, en la sección *Pegar valores*, el primer comando (*Valores*).




- g. Al seleccionarlo, se copiará el valor obtenido por la función, pero no la función. De esa manera, eliminar la Columna de referencia no afectará la operación.
 - i. Presiona ESC, para que el Rango en el portapapeles se libere.
- h. Colócate en F4, que es la Columna que tiene la Función, presiona CTRL+BARRA ESPACIADORA, para que se seleccione la Columna, y luego presiona CTRL+-, para eliminar la columna.
- i. Ahora ya tienes todos los datos en mayúsculas.

Ejercicio 2: Uso de Funciones de grupo de datos.

- ► Tarea 1: Estadística descriptiva usando funciones.
 - 9. Asegúrate de estar en la Hoja **Pruebas**.
 - 10. Selecciona el Rango **G4:G57**.
 - 11. Usando el **Cuadro de nombres**, asígnale al Rango el nombre «Marcas».
 - a. Ahora ya podrás referir todas las marcas obtenidas en maratón, a través de un nombre.
 - 12. Pasa a la Hoja **Resumen**.
 - 13. Determinando la mejor marca.
 - a. Selecciona **B3**.
 - b. Escribe la siguiente fórmula: «=MIN(Marcas)».
 - c. La Función MIN() retorna el valor mínimo en un Rango, mismo que especificaste como argumento.
 - d. Es importante que tomes en cuenta que los datos que tienes en el Rango del ejercicio son horas, por lo que aparece un número que no parece una hora.
 - Acuérdate que Excel maneja las fechas como números decimales, a lo que se conoce como serie.
 - e. Presiona **CTRL+1**, para que aparezca la ventana **Formato de celdas**, y puedas ver el dato en una manera que tenga significado para ti.

f. En la ventana Formato de celdas, selecciona la pestaña Número; en *Categoría*, selecciona «Hora» y en *Tipo*, selecciona la opción que no muestra más que horas, minutos y segundos.



- g. Haz clic en *Aceptar* para aceptar el formato.
- 14. Copiar el formato de una Celda.
 - Si sigues haciendo operaciones con el rango Marcas, lo más probable es que enfrentes el problema del formato.
 - b. Selecciona la celda **B3**, que ya tiene el formato de horas que deseas manejar.
 - c. En la **Cinta de opciones**, selecciona *INICIO Portapapeles Copiar formato*, que es el ícono que aparece con una brocha.

 d. Como puedes ver, el puntero del ratón cambia de forma, indicando que está en modo de copia de formato.

3	Mejor marca:	02:06:32
4	Peor marca:	0.08787037
5	Tiempo promedio:	
6	Diferencia entre el mejor y el peor tiempo:	◆≜

- e. Con el ratón, selecciona el Rango de celdas al que quieres aplicarle el mismo formato que a **B3**.
 - i. En nuestro caso se trata de **B4:B6**.
- f. De esa manera, todos los números de serie que se coloquen en ese rango asumirán el formato hora minuto segundo.
- 15. Determinando la peor marca.
 - a. Selecciona **B4**.
 - b. Escribe la siguiente fórmula: «=MAX(Marcas)».
 - c. La Función MAX() retorna el valor máximo en un Rango, mismo que especificaste como argumento.
- 16. Determinando el tiempo promedio.
 - a. Selecciona **B5**.
 - b. Escribe la siguiente fórmula: «=PROMEDIO(Marcas)».
 - c. La Función **PROMEDIO()** retorna el valor promedio en un Rango, mismo que especificaste como argumento.
- 17. Determinando la diferencia entre el peor y el mejor tiempo.
 - a. Selecciona **B6**.
 - b. En el caso del ejemplo, el número de serie que representa la diferencia entre la mejor marca y la

peor, se obtiene de restarle, al peor tiempo, el mejor tiempo.

c. Escribe la siguiente fórmula: «=B4-B3».

4	А	В
1	Maratón Olímpico - Resumen	
2		
3	Mejor marca:	02:06:32
4	Peor marca:	03:28:53
5	Tiempo promedio:	02:26:15
6	Diferencia entre el mejor y el peor tiempo:	01:22:21


Ejercicio 3: Uso de lógica condicional.

- Tarea 1: Lógica condicional simple, y uso de Insertar función.
 - 18. Asegúrate de estar en la Hoja **Pruebas**.
 - Como puedes ver, en las primeras ediciones de los juegos olímpicos, las mujeres no participaban en la prueba de maratón.
 - a. La tarea que debes realizar es hacer la anotación en la columna *Nota 1*.
 - b. Si no hay dato en la columna *Atleta*, quiere decir que la prueba no fue realizada, y debes colocar la sentencia "No realizada", de lo contrario, se debe colocar la sentencia «Realizada».
 - c. Colócate en la Celda H4.

20. Al lado de la **Barra de fórmulas** se encuentra el ícono **Insertar función**; haz clic en ese ícono.



- d. Presionar **MAYÚS+F3** es equivalente a hacer clic en este ícono.
- 21. Aparecerá la ventana Insertar función.
 - e. En *Buscar una función*, escribe «lógica condicional», dado que quieres conocer una función que te pueda servir para implementar una decisión lógica.
 - f. Haz clic en Ir.
 - g. Te aparecerán en *Seleccionar una función*, aquellas funciones que pudieran servirte. Selecciona «SI».
 - h. Haz clic en Aceptar.



- 22. Si la función requiere argumentos —como es el caso—, Excel abrirá la ventana **Argumentos de función**, en donde podrás proporcionarlos.
 - Si invocas la herramienta Insertar función en una Celda que ya contiene una Función, Excel abrirá directamente la ventana Argumentos de función.




- j. En *Prueba_lógica* se escribe la condición lógica que ha de evaluarse; como el nombre del atleta está en E4, se coloca la expresión «E4=""», que quiere decir «Si E4 es igual a nada, es decir, está vacío».
- k. En *Valor_si_verdadero* es el valor que se desea retornar en caso de que la *Prueba_lógica* sea verdadera. En nuestro caso, si la prueba lógica es verdadera, quiere decir que el nombre del atleta está vacío, por lo cual, ahí escribimos «No realizada».
- En Valor_si_falso es el valor que se desea retornar en caso de que la Prueba_lógica sea falsa. En nuestro caso, si la prueba lógica es falsa, quiere decir que el nombre del atleta está ocupado, por lo cual, ahí escribimos «Realizada».

- m. Mira cómo la ventana Argumentos de función proporciona información de la manera en que se resolverá la función con los argumentos proporcionados.
- n. Haz clic en Aceptar.
- 23. Reproduce la fórmula para todas las Filas en el conjunto de datos.
 - o. Haz doble clic en el ícono de Autollenado de la Celda **H4**, para provocar el autollenado vertical.
 - a. Comprueba que se ha actualizado la información correctamente.


Tarea 2: Lógica condicional anidada y uso de Intellisense.

- 2. Ahora la tarea a realizar es indicar cuál es la mejor marca, y cuál es la peor, informando de ello en la columna *Nota 2*.
 - a. Recuerda que la mejor marca y la peor marca ya las tienes en la Hoja **Resumen**.
 - b. Colócate en I4.
 - c. Comienza a escribir la siguiente fórmula:



d. Mientras escribes, aparecerá la herramienta Intellisense. Cuando veas que está seleccionada la función SI() en Intellisense, presiona TAB, para que se escriba correctamente el nombre de función.


24. Aparecerá la herramienta **Ayuda contextual**, que muestra los argumentos obligatorios que es necesario introducir para que la función esté correcta.



- 25. Proporciona los argumentos de la Función:
 - a. Cuando Excel esté esperando que introduzcas el argumento *prueba_lógica*, escribe «G4=Resumen!\$B\$3», lo que puede leerse como "Si G4 es igual a la referencia absoluta de B3 en la Hoja Resumen".
 - b. Escribe una coma «,» para ir al siguiente argumento.
 - c. Escribe «Mejor marca» como *valor_si_verdadero*.
 - d. Escribe una coma «,» para ir al siguiente argumento.
 - e. Escribe «""» como valor_si_falso.
 - f. Cierra la función con paréntesis.
 - g. Al final, quedaría como sigue el contenido de I4.

- 26. Haz doble clic sobre el ícono de **Autollenado** de **I4** para provocar el autollenado vertical.
 - a. Comprueba que se ha colocado la leyenda de mejor marca a SAMUEL WANJIRU, en Pekín 2008, que se encuentra en la fila 54, con el *ID del evento* 2008H.

- 27. Queremos que en la columna *Nota 2* no solo reporte la mejor marca, sino también la peor.
 - La manera de hacerlo es remplazar el argumento valor_si_falso por una nueva función condicional, de tal manera que se represente una estructura como esta:



b. Al final, I4 debe quedar como sigue:

```
« =SI(G4=Resumen!$B$3,"Mejor marca", SI(G4=Resumen!$B$4,"Peor marca", ""))»
```

- 28. Haz doble clic sobre el ícono de **Autollenado** de **I4** para provocar el autollenado vertical.
 - a. Comprueba que además de señalar a la mejor marca, ahora también se está señalando la peor marca, que es la de THOMAS HICKS en San Luis 1904, que está en la Fila 8 con el *ID de evento* 1904H.

Tarea 4: Funciones de grupo con condicional.

- 29. Ahora el trabajo por hacer es contar el número de triunfos que tiene cada país, por lo que deberemos contar qué tantas veces aparece el nombre de país en los datos.
- 30. Asegúrese de estar en la Hoja Pruebas.
- 31. Selecciona el Rango H4:H57.
- 32. Usando el **Cuadro de nombres**, asigna al Rango el nombre «Países».
- 33. Ahora selecciona la Hoja **Resumen**.
- 34. Selecciona la Celda **B9**.
- 35. Presiona MAYÚS+F3 para que aparezca la ventana Insertar función.
 - a. En *Buscar una función* escribe «Contar», y haz clic en *Ir*.
 - b. Selecciona «CONTAR.SI», y haz clic en *Aceptar*.
 - c. Aparecerá la ventana **Argumentos de función**.
 - d. En *Rango* escribe «Países», que es el Rango donde se buscarán coincidencias para contar.
 - e. En *Criterio* escriba «A9».
 - f. La función queda como sigue:

«=CONTAR.SI(Países, A9)»

g. Dado que en A9 tenemos el dato «ALEMANIA ORIENTAL (GDR)» la función equivale a decir «cuenta cuántas veces tenemos en el Rango Países el valor de "ALEMANIA ORIENTAL (GDR)"».

- 36. Haz doble clic sobre el ícono de **Autollenado** de **B9** para provocar el autollenado vertical.
 - Comprueba que se ha determinado correctamente el número de veces que cada país ha ganado la prueba de Maratón.

Tarea 5: Uso de Análisis rápido.

- 37. Ahora el trabajo consiste en comprobar el uso de la herramienta de **Análisis rápido**, que facilita algunas tareas de formato y de aplicación de Funciones de grupo.
- 38. Imagina que deseas saber el número de pruebas de maratón que se han realizado en toda la historia. Para conocer ese dato, tendrías que sumar todas las medallas de oro que han obtenido todos los países en la historia.
 - a. Selecciona el Rango B9:B25.
 - b. Como podrás ver, en la parte inferior derecha aparece el ícono de Análisis rápido.

	А	В	
8	Países	Medallas Oro	
9	ALEMANIA ORIENTAL (GDR)	2	
10	ARGENTINA (ARG)	2	
11	CHECOSLOVAQUIA (TCH)	1	
12	COREA DEL SUR (KOR)	1	
13	EQUIPO UNIFICADO (EUN)	1	
14	ESTADOS UNIDOS (USA)	4	
15	ETIOPÍA (ETH)	6	
16	FINLANDIA (FIN)	2	
17	FRANCIA (FRA)	3	
18	GRECIA (GRE)	1	
19	ITALIA (ITA)	2	
20	JAPÓN (JPN)	3	
21	KENIA (KEN)	1	
22	PORTUGAL (POR)	2	
23	RUMANIA (ROU)	1	
24	SUDÁFRICA (RSA)	2	
25	UGANDA (UGA)	1	
26			/
27			73

- c. Haz clic en el ícono de **Análisis rápido**.
- d. Aparecerá la herramienta, que te permite seleccionar diferentes comandos.
- e. Selecciona primero alguno de los grupos de comandos que aparecen en la parte superior en forma de pestañas.

 Selecciona **TOTALES**.
- f. Luego selecciona alguno de los comandos disponibles en la pestaña. Selecciona *Sumar*.



g. Verás cómo se ha agregado de manera automática la sumatoria.

Tarea 5: Contando blancos y construir una sentencia textual con información.

- 39. Selecciona la Hoja **Pruebas**.
- 40. Imagina que quieres agregar una leyenda que haga notar las veces en que la prueba femenil no se realizó.
- 41. La forma de saber ese dato es contando las veces en que el nombre del atleta está vacío, es decir, que está en blanco.
 - a. Toma en cuenta que los nombres de los atletas se encuentran en el Rango **E4:E57**.
- 42. Selecciona la Celda A59.
- 43. Escribe la siguiente fórmula:

«=CONTAR.BLANCOS(E4:E57)»

- 44. La Función **CONTAR.BLANCOS()** contará las celdas en blanco del Rango especificado. En este caso, son 19.
- 45. Imagina que no quieres mostrar el número, sino que quieres una leyenda que diga:
 - «La prueba femenil no se realizó en 19 ocasiones.>>
- 46. Como notarás, esa leyenda puede construir si concatenas los siguientes elementos: «La prueba femenil no se realizó en », «CONTAR.BLANCOS(E4:E57)», y « ocasiones».
- 47. Colócate en **A59** y presiona **F2**, para entrar en modo edición.
- 48. Modifica la fórmula, de tal manera que quede de la siguiente manera:

≪=CONCATENAR("La prueba femenil no se realizó en ",CONTAR.BLANCO(E4:E57)," ocasiones.")>>

49. Comprueba que se ha agregado la leyenda como la queríamos.

Ejercicio 4: Uso de BUSCARV().

- Tarea 5: Buscando información a partir de una clave.
 - 50. Ahora el trabajo por hacer es colocar el nombre del atleta y de su país de procedencia, junto a la información de la mejor marca y de la peor.
 - 51. Asegúrate de estar en la Hoja **Pruebas**.
 - 52. Selecciona el Rango A4:J57.

- 53. Usando el **Cuadro de nombres**, asigna al Rango el nombre «Datos».
 - a. Analiza qué hay en cada Columna, considerando que la primera es la 1, la segunda es la 2, y así.
 - b. Toma nota de los *ID de evento* tanto de la mejor marca como de la peros.
- 54. Ahora selecciona la Hoja **Resumen**.
- 55. Para buscar la información del nombre del atleta que obtuvo la mejor marca:
 - c. Recuerda que el atleta que obtuvo la mejor marca tiene el *ID de evento* 2008H.
 - d. Toma en cuenta que los datos están en el Rango denominado *Datos*.
 - e. Recuerda que el nombre del atleta está en la columna 5 de los datos.
 - f. Haremos la búsqueda por el *ID de evento*, que es un identificador único, así que podemos buscar una coincidencia exacta.
- 56. Colócate en C3.
- 57. Escribe «=BUSCARV(», y presiona **MAYÚS+F3**, para invocar la ventana **Argumentos de función**.
 - g. En Valor_buscado escribe «2008H».
 - h. En *Matriz_buscar_en* escribe «Datos».
 - i. En *Indicador_Columnas* escribe «5».
 - j. En *Ordenado* escribe «FALSO».
 - k. Haz clic en *Aceptar*, para que se realice la búsqueda.
 - l. La Función quedó como sigue:

«=BUSCARV("2008H",Datos,5,FALSO)»

58. Comprueba que se ha recuperado el dato correctamente.

- 59. Ahora complementa la Hoja, de tal manera que las siguientes Celdas queden con las siguientes fórmulas.
 - a. Harás uso también de la información de la columna que guarda el nombre de país, que está en la columna 6, de *Datos*:
 - b. En **C4** escribe «=BUSCARV("1904H",Datos,5,FALSO)», para sacar el nombre del atleta con el peor tiempo.
 - c. En **D3** escribe «=BUSCARV("2008H",Datos,6,FALSO)», para sacar el nombre del país del mejor tiempo.
 - d. En **D4** escribe «=BUSCARV("1904H",Datos,6,FALSO)», para sacar el nombre del país con el peor tiempo.
- 60. Guarda tu trabajo presionando CTRL+G.

Ejercicio 5: Expresiones complejas con lógica condicional y manejo de texto.

► Tarea 1: Generando nombres cortos.

- 61. Esta tarea es opcional, y debe realizarse si el tiempo lo permite y si quieres probar tus habilidades al máximo.
- 62. La tarea a realizar es la siguiente: Por alguna razón, en toda la documentación olímpica los nombres se manejan con un máximo de 15 posiciones.
 - a. Tu trabajo es generar un nombre corto en caso de que el nombre del atleta sea mayor a 15 posiciones.
 - b. Si el nombre es menor o igual a 15 posiciones, se queda como está.
 - c. Si es mayor a 15 posiciones, se abreviará el primer nombre, colocando sólo la inicial. Ejemplos:

SPYRIDON LOUIS → SPYRIDON LOUIS → H. KOLEHMAINEN

- 63. Lo primero que tienes que hacer es encontrar qué nombres son mayores a 15 posiciones.
 - a. Colócate en la Celda **J4**, que es la *Nota 3*.
 - b. Toma nota que el nombre del atleta para esa Fila está en la Celda **E4**.
 - c. Para saber el largo en caracteres del nombre del atleta para esa Fila—E4—, tienes que utilizar la Función LARGO(). En J4 escribe la siguiente fórmula:

«=LARGO(E4)»

- d. Haz doble clic en el ícono de **Autollenado** de la Celda **J4**, para que se realice el autollenado vertical.
- e. Como puedes notar, hay varios nombres que exceden el largo de 15 caracteres.
- 64. Lo que sigue por hacer es colocar el nombre del atleta si es que no excede de 15 posiciones, y por lo pronto, hacer notar qué nombres si requieren ser modificados.
 - a. Edita la fórmula que está en J4, de tal manera que cuando el largo del nombre del atleta —que está en E4—sea mayor a 15, se retorne un valor de «MODIFICAR», o de lo contrario, que el nombre se quede como está. Desde luego, utilizarás la Función SI(), de la siguiente forma:

«=SI(LARGO(E4)>15,"MODIFICAR",E4)»

- Esto se lee de la siguiente forma «Si el largo del texto contenido en E4 es mayor a 15, retorna "MODIFICAR", y de lo contrario, retorna el valor que está en E4».
- c. Haz doble clic en el ícono de **Autollenado** de la Celda **J4**, para que se realice el autollenado vertical.
- 65. Ahora la cosa se pone un poco más complicara, pues en lugar del texto «MODIFICAR», debemos colocar el nombre corto, lo que implica varias acciones.
 - a. El nombre corto es la concatenación de varios elementos, por lo que utilizaremos la Función **CONCATENAR()**.
 - El primer elemento del nombre corto es la primera inicial. Para obtenerla, es necesario utilizar la Función EXTRAE(), de la siguiente forma:

«EXTRAE(E4,1,1)»


- c. Lo que se puede leer como «Extrae del texto que está en E4, a partir del caracter 1, la cantidad de 1 caracter».
- d. El segundo elemento del nombre corto es un punto, más un espacio: «. »
- e. Complementa la fórmula que está en **J4**, para que vayas viendo los resultados. Hasta el momento, está así:
 - «=SI(LARGO(E4)>15,CONCATENAR(EXTRAE(E4,1,1),". "),E4)»
- f. Haz doble clic en el ícono de **Autollenado** de la Celda **J4**, para que se realice el autollenado vertical.



- g. El siguiente elemento a concatenar es el resto del nombre del atleta, a partir del primer espacio en blanco que se encuentre.
- h. Para saber dónde se encuentra el primer espacio en blanco, debemos utilizar la función ENCONTRAR(), con la siguiente sintaxis:

«ENCONTRAR(" ",E4,1)»


- Lo que se puede leer como «Encuentra un espacio en blanco " " en E4, buscando desde la posición 1 hacia adelante».
- j. Toma en cuenta que ENCONTRAR() te dirá la posición en la que se encuentra el primer espacio, pero tú requieres extraer de donde está el espacio uno más (+1).
- k. El resto del nombre corto es igual a extraer de E4, a partir de donde se encuentra el primer espacio en blanco, +1, una cantidad de caracteres igual al largo de E4 menos la posición donde se encontró el primer espacio.



l. Se agrega el resto del nombre a la concatenación, y la fórmula quedaría finalmente como sigue:

```
«=SI(LARGO(E4)>15, CONCATENAR(EXTRAE(E4,1,1), ". ",
EXTRAE(E4, ENCONTRAR(" ",E4,1)+1, LARGO(E4)-ENCONTRAR(" ",E4,1))),E4)»
```

m. Haz doble clic en el ícono de **Autollenado** de la Celda **J4**, para que se realice el autollenado vertical.



- n. Comprueba que los nombres que estaban con una longitud mayor a 15, fueron abreviados satisfactoriamente.
- o. Guarda el trabajo, presionando CTRL+G.

www.Aprenda.mx :: info@aprenda.mx :: 01-800-APRENDA :: México

Material complementario al curso en línea ¡Aprenda Excel Ya! Su uso en otro contexto está prohibido.

